# $Id: PKGBUILD 150349 2012-02-17 07:28:10Z allan $
# Maintainer: Dale Blount <dale@archlinux.org>

pkgname=perl-sys-hostname-long
_realname=Sys-Hostname-Long
pkgver=1.4
pkgrel=5
pkgdesc="Perl module getting a full hostname"
arch=('any')
license=('PerlArtistic')
url="http://search.cpan.org/dist/${_realname}/"
depends=('perl>=5.10.0')
options=(!emptydirs)
source=(http://search.cpan.org/CPAN/authors/id/S/SC/SCOTT/${_realname}-${pkgver}.tar.gz)
md5sums=('a5b3c23754cbcff10e283d16c42ec3d3')

build() {
  cd ${srcdir}/${_realname}-${pkgver}
  # install module in vendor directories.
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  cd ${srcdir}/${_realname}-${pkgver}
  make install DESTDIR=${pkgdir}
}
